A Spacecraft Simulator

Gary Sivak
19 Madison Trail
Hopatcong NJ 07843

This article describes a BASIC program that enables the user to design and put into orbit a multistage spacecraft launched from Earth-based conditions. By asking for engine throttle settings, thrust angles, and firing times, your computer puts you at the controls of a multistage spacecraft of your own design as you pilot it from the Earth's surface into orbit. Continuous data displays of the user's status after each maneuver are presented, as well as arrays of altitude and range information for possible plotting at the end of the mission. The following is a description of the program operation.

The program first asks for and verifies all ship design parameters, the first being the number of stages. Then the iteration time (dt) in seconds and the height in miles of the desired orbit are required. During each iteration, the computer calculates formulas of the form:

$$V_{final} = V_{initial} + \text{acceleration} \times dt \quad (1)$$

The final values are then taken as the initial ones for the next iteration. An iteration time evenly divisible into one second is recommended; 0.1 seconds is suggested for faster than real-time computation. A figure of 0.01 seconds, for example, will give a slightly better mathematical accuracy but at the expense of ten times more processing time.

The craft is assembled from top down, the weight of the payload in

Text continued on page 108

Listing 1: *BASIC listing of the rocket launcher program.*

```
                    ROCKET LAUNCHER PROGRAM

10 DIM A(100),A0(100),A1(7),A2(7),A3(6),A4(6)
20 PRINT "DESIGN AND ORBIT A SPACE SHIP. TYPE NO. STAGES UP TO 6. "
30 INPUT A5
40 PRINT "VERIFICATION, ";A5;" STAGES."
50 A6 = A5 + 1
60 PRINT "ENTER ITERATION TIME IN SEC., AND ORBIT HEIGHT IN MI. "
70 PRINT ".1 SEC. IS OK AND .01 BETTER, BUT WITH MORE CPU TIME. "
80 INPUT A7,A8
90 PRINT "VERIFICATION, ITERATION TIME ";A7;", ORBIT HEIGHT ";A8
100 PRINT "ENTER PAYLOAD WEIGHT IN POUNDS. "
110 INPUT A2(A6)
120 A1(A6) = 0.0
130 PRINT "VERIFICATION, PAYLOAD WEIGHT, ";A2(A6)
140 FOR A9 = 1 TO A5
150 B = A6 - A9
160 B0 = B + 1
170 PRINT "ENTER STAGE ";B;" FUEL AND HULL WEIGHTS IN LBS. "
180 INPUT A1(B), A2(B)
190 PRINT "STAGE ";B;" FUEL ";A1(B);" LBS., HULL ";A2(B);" LBS. "
200 A2(B) = A2(B) + A2(B0) + A1(B0)
210 B1 = A2(B) + A1(B)
220 PRINT "ENTER STAGE ";B;" THRUST AT LEAST ";B1;" .LBS. "
230 INPUT A3(B)
240 PRINT "STAGE ";B;" THRUST, ";A3(B);" LBS. "
250 PRINT "ENTER SPECIFIC IMPULSE OF STAGE ";B;" FUEL/OXIDIZER. "
260 PRINT "THIS IS THE THRUST-TO-BURN RATE RATIO. "
270 PRINT "FOR GASOLINE =250, PEROXIDE =300, LIQUID HYDROGEN =500. "
280 INPUT A4(B)
290 PRINT "VERIFICATION, STAGE ";B;" SPECIFIC IMPULSE ";A4(B)
300 NEXT A9
310 B2 = 10
320 B3 = B2 * A7
330 B4 = 360
340 B5 = B3 / 100.0
350 B6 = 5280. * .3048
360 B7 = 6.67E-11 * 5.983E24
370 B8 = ATN(1.) / 45.
380 B9 = 90.
390 C = 1.0
400 C0 = SQR(B7/9.80665)
410 C1 = C0
420 C2 = SQR(B7/(C0+B6*A8)) / .3048
430 C3 = 0.0
440 C4 = 0.0
450 C5 = 0.0
460 C6 = 0.0
470 C7 = 0.0
480 C8 = 0.0
```

Listing 1 continued on page 108

Circle 349 on inquiry card.

Listing 1 continued:

```
490 C9 = 0.0
500 D = 0.0
510 D0 = 0.0
520 D1 = 0.0
530 D2 = 0.0
540 D3 = 0.0
550 PRINT "THE SHIP CAN SWIVEL ";B2;" DEG/SEC. "
560 PRINT "EARTH'S GRAVITY IS 32.174 FT/SEC/SEC. "
570 PRINT "FORWARD VELOCITY NEEDED FOR ORBIT ";C2;" FT/SEC. "
580 D = D + 1
590 D4 = A2(D) / 2.2046
600 D5 = A3(D) / A4(D) / 2.2046
610 D6 = A1(D) / 2.2046
620 D7 = D6
630 D8 = A3(D)/2.2046*9.80665
640 PRINT "IGNITION OF STAGE ";D;", ENTER THE STAGE NUMBER. "
645 INPUT X1
650 GO TO 1090
660 PRINT "ENTER THROTTLE SETTING IN %, FROM 0 TO 100, "
670 PRINT "THRUST ANGLE IN DEG. FROM -";B4;" TO ";B4
680 PRINT "AND BURN TIME IN SECONDS. "
690 INPUT D9, E, E0
700 D9 = ABS(D9 / 100.0)
710 E1 = D9 * D8
720 E2 = D9 * D5 * A7
730 E3 = E2 / 100.
740 E4 = E0 - (A7 / 100.0 )
750 E5 = C5 * C1
760 E6 = 0.0
770 IF E0 = 0.0 THEN 1080
780 IF C1 < C0 THEN 1080
790 E6 = E6 + A7
800 E7 = D7 - E2
810 E8 = E1 / (D4 + (D7 + E7 ) / 2.0 )
820 IF E7 >= E3 THEN 850
830 E7 = 0.0
840 E8 = 0.0
850 IF ABS( E - B9 ) < B5 THEN 930
860 IF E < B9 THEN 890
```

Listing 1 continued on page 110

Text continued:

pounds being required first. For each stage, the computer then asks for the weights of the fuel and hull (or tanks), the maximum thrust desired, and the specific impulse of the fuel. To insure the possibility of achieving orbit, a fuel to hull weight ratio of 4 or 5 to 1 is suggested. A thrust of about 20 percent more than the minimum amount required to lift the ship is suggested, so that the ship has sufficient acceleration, even when heavily laden with fuel.

Specific impulse is a figure of merit for fuel performance, the thrust to burn-rate ratio. Suggested values for different fuels are given in the program. Knowing the thrust and specific impulse defines the burn rate, and knowing the amount of fuel on board designates how long it will last at full throttle expenditure. Next, a printout chart, to be described shortly, displays initial fuel, altitude, and the velocity status of the ship.

At this point, the flight begins; the user is in control, and must specify the throttle setting, firing angle, and burn time for each maneuver. The force on the ship (in newtons) is first computed from the throttle setting

Listing 1 continued:

```
870 B9 = B9 + B3
880 GO TO 900
890 B9 = B9 - B3
900 E9 = B9  * B8
910 C4 = COS(E9)
920 C = SIN(E9)
930 F = E8 * C4
940 F0 = E8 * C
950 F1 = C5 + F * A7
960 C6 = ( C5 + F1 ) / 2.0
970 C7 = C7 + C6 * A7
980 F2 = F0 + C6**2 / C1  - B7 / C1**2
990 F3 = C8 + F2 * A7
1000 F4 = C1 + ( C8 + F3 ) / 2.0 * A7
1010 IF D9 <> 0.0 THEN 1030
1020 F1 = E5 / F4
1030 D7 = E7
1040 C5 = F1
1050 C8 = F3
1060 C1 = F4
1070 IF E6 < E4 THEN  770
1080 C3 = C3 +  E6
1090 D2 = D2 + 1
1100 A(D2) = ( C1 - C0 ) / .3048
1110 IF C9 >= A(D2) THEN 1130
1120 C9 =  A(D2)
1130 IF A(D2) >= 0.0  THEN 1150
1140 A(D2) = 0.0
1150 IF A(D2) < 400000.0  THEN 1170
1160 D3 = D3 + 1
1170 F5 = A(D2) / 5280.
1180 F6 = C8 / .3048
1190 F7 = F6 * 15./22.
1200 F8 = C5 / .3048
1210 F9 = F8 * 15./22.
1220 A0(D2) =  C7 / B6
1230 G = 100. * D7 / D6
1240 G0 = D7 / D5
1250 G1 = B7 / C1**2 - C6**2 / C1
1260 G2 = D8 / (D4 + D7) / .3048
1270 G3 = G2 * 15. / 22.
1280 G4 = G2 - ( G1 / .3048 )
1290 G5 = G4 * 15. / 22.
1300 G6 = G1 / .3048 / G2
1310 G7 = 100. * G6
1320 G8 = 90.0
1330 IF G6 >= 1.0 THEN 1350
1340 G8 = ATN( G6 / SQR( 1.0 - G6**2 ) ) / B8
1350 G9 = SQR( B7 / C1 ) / .3048
1360 H = 100. * F8 / C2
1370 H0 = 100. * A(D2) / ( A8 * 5280. )
1380 H1 = 100. * F8 / G9
1390 H2 = ( C2 - F8 ) / G2
1400 H3 = ( G9 - F8 ) / G2
1410 IF F6 = 0.0 THEN 1440
1420 H4 =  (A8*5280. - A(D2) ) / F6
1430 IF H4 <= 9999.99 THEN 1460
1440 H4 = 9999.99
1450 REM-TIMES OVER 9999.99 SET TO 9999.99 TO NOT EXCEED DISPLAY.
1460 IF D3 <> 1.0 THEN 1480
1470 PRINT "400K FT. ACHIEVED, YOU ARE IN VACUUM. "
1480 PRINT "FLIGHT TIME","FUEL LEFT","AT FULL THROT.","SHIP ANGLE"
1490 PRINT C3;"SEC,",G;"%",G0;"SEC,",B9;"DEG."
1500 PRINT "  "
1510 PRINT "ALTITUDE","ASCENT RATE","FORWARD V.","RANGE"
1520 PRINT A(D2);"FT.",F6;"FT/SEC",F8;"FT/SEC",A0(D2);"MI."
1530 PRINT F5;"MI.",F7;"MI/HR.",F9;"MI/HR."
1540 PRINT "  "
1550 PRINT "MAX ACCEL","MAX VERT ACCEL","ANGLE(C.A.)","THROT(C.A.)"
1560REM-ANGLE(C.A.), CRITICAL ANGLE FOR CONST. ASCENT AT FULL THROT.
1570REM-THROT(C.A.), CRITICAL THROT. OF CONST. ASCENT AT 90DEG.
1580 PRINT G2;"FT/S/S",G4;"FT/S/S","FULL THROT.","VERT. POS."
1590 PRINT G3;"MI/H/S",G5;"MI/H/S",G8;"DEG.",G7;"%"
1600 PRINT "  "
1610 PRINT H;"% ORBITAL VELOCITY",H0;"% ORBITAL HEIGHT."
1620 PRINT H1;"% VELOCITY NEEDED FOR ORBIT AT CURRENT ALTITUDE."
1630 PRINT "  "
1640 PRINT " "," ","TIME TO ACHIEVE:"
1650 PRINT "ORB. ALT.","ORB. VEL.","CUR. ALT. ORB. VEL."
1660 PRINT "AT CUR. RATE","AT FULL THROT.","AT FULL THROT."
1670 PRINT H4;"SEC.",H2;"SEC.",H3;"SEC."
1680 PRINT "  "
```

Listing 1 continued on page 111

and maximum specified thrust. Also, note that a firing angle of ninety degrees is vertically upward, and angles less than ninety degrees are to the right, or east, etc. A one hundred percent throttle setting at ninety degrees for fifteen or twenty seconds is suggested to gain altitude before beginning to swivel the ship to achieve horizontal orbital velocity.

The amount of fuel used during an iteration is simply the throttle setting, times the maximum burn rate, times dt. This amount, subtracted from the weight of the fuel at the beginning of an iteration, gives the amount remaining at the end. The amount of fuel available during an iteration is taken as the average of the amounts before and after. This is added to the weight of the tanks and the upper stages that the engines must lift, and is the instantaneous weight (in kilograms) of the craft. Dividing into the thrust force yields the current engine thrust acceleration A_t during the iteration, in meters per second per second (m/s²).

For a given firing angle, the horizontal and vertical components of this acceleration, a_{th} and a_{tv}, are taken. Horizontal velocities and the range are computed by:

$$V_{fh} = V_{ih} + a_{th} \times dt \qquad (2)$$

$$V_{avh} = (V_{ih} + V_{fh})/2 \qquad (3)$$

$$\text{range} = \text{range} + V_{avh} \times dt \qquad (4)$$

where, for a particular iteration, V_{ih} is the initial horizontal velocity, V_{fh} is the final horizontal velocity, and V_{avh} is the average of the two.

The total outward vertical acceleration a_{rv} is computed by adding centrifugal acceleration to the engine acceleration and subtracting gravity's downward contribution as follows:

$$a_{rv} = a_{tv} + (V^2_{avh}/r_{iv}) - GM/r^2_{iv} \qquad (5)$$

where, r_{iv} is the initial value of the vertical distance of the ship from the Earth's center, G is the gravitational constant, and M is the mass of the Earth. From the vertical acceleration, the velocities and altitude are computed just as the horizontal components were computed in equations 2 thru 4.

From physics, it will be noted that if no external force is applied by the engines, the rocket's angular momentum is a constant. For each maneuver, therefore, the computer retains

The following constants were used in listing 1:

G: Gravitational constant,
$6.67 \times 10^{-11} Nm^2/kg^2$
M: Mass of the earth,
$5.983 \times 10^{24} kg$
g: Gravitational acceleration,
9.80665 N/kg,
$m/sec^2 = 32.174\ ft/sec^2$

0.3048 meters/foot
2.2046 pounds/kg

Listing 1 continued:

```
1690 IF H < 100.0 THEN 1760
1700 IF H0 < 100.0 THEN 1760
1710 D0 = D0 + 1
1720 IF D0 > 1 THEN 1760
1730 PRINT "IN DESIRED ORBIT.  TO CONTINUE ENTER 1, TO PLOT ENTER 2. "
1740 INPUT H5
1750 IF H5 = 2 THEN 1920
1760 IF C3 = 0.0 THEN 660
1770 IF D7 <= E3  THEN 1800
1780 IF A(D2) <= 0.0 THEN  1800
1790 GO TO 660
1800 IF A(D2) = 0.0  THEN 1890
1810 IF D < A5 THEN  580
1820 D1 = D1 + 1
1830 IF D1 <> 1 THEN  1850
1840 PRINT "LAST STAGE SHUTDOWN."
1850 IF D0 <> 0.0 THEN 1880
1860 IF A(D2) <= 0.0 THEN 1880
1870 GO TO 660
1880 IF A(D2) > 0.0 THEN 1920
1890 H6 = INT( SQR( F6**2 + F8**2 ) + .5)
1900 H7 = INT( SQR( F7**2 + F9**2 ) + .5)
1910 PRINT "YOU CRASHED AT ";H6;" FT/SEC, ";H7;" MI/HR. "
1920 PRINT "AFTER ";D2;" PLOT POINTS: "
1930 FOR H8 = 1 TO D2
1940REM-PLOT A(H8) Y-AXIS, VS. A0(H8) X-AXIS, ALTITUDE VS. RANGE.
1950 NEXT H8
1960 H9 = 25.0
1970 REM-LOWER 25% CUTOFF OF ALTITUDE FOR A BLOWUP PLOT.
1980 I = C9 * H9 / 100.0 * 1.0001
1990 I0 = D2 + 1
2000 I0 = I0 - 1
2010 IF A(I0) > I THEN 2000
2020 I1 = 100.0 * A0(I0) / A0(D2)
2030 PRINT "LOWER ";H9;"% OR ";I;" MI. OF MAX ALT. ATTAINED."
2040 PRINT "FIRST ";I1;"% OR ";A0(I0);" MI. OF TOTAL RANGE."
2050 PRINT "WITH ";I0;" STEPS:"
2060 FOR I2 = 1 TO I0
2070 REM-PLOT A(I2) Y-AXIS, VS. A0(I2) X-AXIS, LOWER ALT. VS. RANGE."
2080 NEXT I2
2090 END
```

the product of horizontal velocity and distance from the Earth's center. If the engines are off during an iteration, the new horizontal velocity is set equal to this product divided by the new vertical distance value at the end of the iteration. Thus, angular momentum is conserved. As the ship coasts towards Earth, its horizontal velocity increases slightly, and would decrease slightly if the ship were receding. Quantities are then reinitialized and the next iteration begins.

When a firing sequence is completed, an important quantity Q is computed. It is the ratio of the net downward acceleration (gravitational minus centrifugal) to the total acceleration. The engines can currently deliver:

$$Q = \left(\frac{GM}{r_{iv}^2} - \frac{V^2_{avh}}{r_{iv}} \right) / a_t \quad (6)$$

Multiplied by 100, this is the critical throttle setting which will cause the ship to hover if stationary, or move vertically at a constant speed without accelerating. It is also the sine of the critical angle of ascent at which the vertical component of thrust equals the current weight of the ship. The angle, equal to the inverse sine of Q is alternatively computed from:

$$\text{angle} = \tan^{-1} (Q/\sqrt{1.0 - Q^2})$$

At this time, distance and velocity values are converted from metric to English units for display purposes.

The first information printed consists of the elapsed flight time, the current ship angle, and the fuel left, both as a percentage of the original amount, and the number of seconds left at full throttle. Next, the program prints the altitude in miles and feet, the ascent rate and forward velocity in miles per hour and feet per second, and the number of miles down range.

The next printed information consists of the critical angle and throttle values of constant ascent, the maximum acceleration the engines can deliver, and the maximum vertical acceleration against gravity in both miles per hour per second and feet per second^2. For example, if the engine can deliver about 40ft/s^2 the ship can accelerate at 8ft/s^2 against gravity.

Next the percentages of the orbital velocity and altitude are presented. The final items displayed are the time to achieve orbital altitude at the current ascent rate, and the time to achieve orbital velocity at the current full throttle rate of horizontal acceleration.

At this point the user is ready for the next move, and must again specify a new throttle setting, firing angle, and burn time. Finally, at the end of the mission (either when you achieve orbit, or run out of fuel), you can plot a picture of your trajectory, altitude versus range, and an expanded plot of the start of your mission, the lower 25 percent of your total attained altitude.

Have fun. As you will soon learn, getting your spacecraft to achieve orbit is no easy task.■